1 Bevezetés, alap számológép műveletek

A sor végén SHIFT+ENTER (vagy a numerikus billentyűzeten lévő ENTER) lenyomásával értékeljük ki a sort.
A beviteli sor beírása után automatikusan ; jelet tesz a sor végére SHIFT+ENTER leütésénél, ami azt jelenti, hogy ki is értékeli a sort és ki is írja az eredményeket.

```
(%i1) 17+7;
(%o1) 24

(%i2) 333−7;
(%o2) 326

(%i3) 7/9;
```

(%o3) $\frac{7}{9}$

```
(%i4) 11·22;
(%o4) 242

(%i5) 7!;
(%o5) 5040

(%i6) (5e−3)·100;
(%o6) 0.5

(%i7) 9^2;
(%o7) 81
```

Vegyük észre, hogy mind a bemeneti input sor, mind a kimeneti output sor címkéket kapott, pl %i1 és %o1. Ezekre mint változókra hivatkozhatunk a későbbiekben. Például nézzük meg a %o1 / %o4 értékét:

```
(%i8) %o1+%o4;
(%o8) 266
```

Amennyiben szükséges, definiálhatunk mi is saját változókat a : jellel. De fontos, hogy ne használjunk védett változó neveket. Ha a sor végére $ jelet rakunk akkor kiértékeli a sort, de nem írja ki az eredményt. Példa:

```
(%i11) a:5$
       b:8$
       c:10$

(%i13) (a+b·c)^a;
       (a+b·c)··a;
(%o12) 4437053125
(%o13) 4437053125
```

Érdemes megjegyezni, hogy a dupla szorzás jel hatványozást jelent. Sokszor kényelmesebb ezt használni.

Ha egy változó értékét törölni akarjuk akkor használhatjuk a kill() parancsot. Példa:

```
(%i14) kill(a);
(%o14) done

(%i17) a;
       b;
       c;
(%o15) a
(%o16) 8
(%o17) 10
```

Láthatjuk, hogy a értéke törlődött. kill(all) paranccsal pedig minden változót törölhetünk.

```
(%i18) kill(all);
(%o0) done
```

Vegyük észre, hogy a szoftver különböző színeket alkalmaz a struktúrában. A beviteli értékek zöldek vagy barnák, attól függően, hogy számértékről vagy változóról van szó, míg a kimeneti értékek feketék. Emellett kék az egyéb karaktereke és szintén barnák a beépített függvények.

Néhány egyszerű alapvető parancs:

```
(%i1) sqrt(5);
```

(%o1) $\sqrt{5}$

```
(%i2) sqrt(5.0);
(%o2) 2.23606797749979
```

Vegyük észre, hogy a program az sqrt(5)-t nem értékeli ki numerikusan, hanem szimbolikusan kezeli. De ha az 5 értékét lebegőpontosan 5.0-ként adjuk meg akkor az eredmény is lebegőpontos lesz. Ez a többi függvényre is igaz. De ha szükséges akkor megadhatjuk, hogy mindenképpen lebegőpontos formában kérjük az eredményt a float() paranccsal:

```
(%i4)  sqrt(5);
       float(%);
(%o3)  √5
(%o4)  2.23606797749979
```

A fenti megadásnál a % változó az előzőleg kiértékelt eredményre utal, jelen esetben az sqrt(5)-re. Sok esetben célszerű használni.
Néhány további alapvető függvény:

```
(%i5)  log(10.0);
(%o5)  2.302585092994046
```

Látható, hogy a log függvény nem 10-es alapú hanem a természetes logaritmust adja. Ha más alapú logaritmus kell, akkor célszerű saját függvényt definiálni hozzá a := jellel. Példaképpen a legyen log10() a 10-es alapú logaritmust megadó függvény:

```
(%i6)  log10(x):=log(x)/log(10.0)$

(%i7)  log10(10.0);
(%o7)  1.0
```

Ez alapján látható, hogy definiálhatunk saját függvényeket viszonylag egyszerűen. Például:

```
(%i8)  fgv1(x):=float(x·x−100·x+200)$

(%i9)  fgv1(−7);
(%o9)  949.0
```

Néhány további alapvető függvény:

```
(%i11) cos(60);
       float(%);
(%o10) cos(60)
(%o11) −0.9524129804151563
```

Fontos, hogy a szögfüggvények argumentumát radiánban kell megadni! Ha mi fokban kívánjuk beírni akkor át kell váltani fokba a radiánt. De ha sokszor kell alkalmazni szögfüggvényeket akkor akár célszerű lehet külön definiálni 1 fok értékét radiánban. Például:

```
(%i12) fok:%pi/180$

(%i13) cos(60·fok);
```

(%o13) $\frac{1}{2}$

Fent látható, hogy a %pi egy védett változó, amihez a Pi értéke van rendelve. Hasonló védett változó a %e, ami az Euler-féle szám, valamint %i jelenti a képzetes egységet. %phi pedig az aranymetszés. Példa:

```
(%i18) float(%pi);
       float(%e);
       float(%phi);
       sqrt(−1);
       %i·%i;
(%o14) 3.141592653589793
(%o15) 2.718281828459045
(%o16) 1.618033988749895
(%o17) %i
(%o18) −1
```

A lebegőpontos számábrázolás köztudottan kerekítési hibákhoz vezethet. Amennyiben nagyobb pontosságú számábrázolásra van szükségünk akkor használhatjuk a "big float" formátumot, amikor kontrollálhatjuk a számábrázolásban a digit-ek számát. Ilyenkor a "scientific form" ábrázolásnál b-t kell használnunk e helyett. Utóbbi esetben "big float" formátumban kerül tárolásra az érték. A tárolandó digit-ek számát az fpprec belső változó tartalmazza, aminek default értéke 16, de ezt felülírhatjuk. Példaként nézzük meg a Pi értékét 50 esetben:

```
(%i19) float(%pi);
(%o19) 3.141592653589793

(%i21) fpprec:50$
       bfloat(%pi);
(%o21) 3.1415926535897932384626433832795028841971693993751b0

(%i22) float(10/7);
(%o22) 1.428571428571429
```

```
(%i23) bfloat(10/7);
(%o23) 1.428571428571428571428571428571428571428571428571b0
```

Egy és többváltozós függvény definiálása:

```
(%i24) f(x) := x^2$

(%i25) f(7.7);
(%o25) 59.29

(%i26) g(x,y) := x^2+exp(10.0·y)$

(%i27) g(8,2);
```

(%o27) $4.851652594097903\ 10^{8}$

2 Algebrai műveletek

Sok esetben egy kifejezés értékébe szükséges behelyettesíteni. Ezt megtehetjük a subst() paranccsal. Példa:

```
(%i28) F:m·a;
(F)    a m

(%i29) subst(a=7,F);
(%o29) 7 m
```

Ezzel F értéke, definíciója nem változik:

```
(%i30) F;
(%o30) a m
```

Azonos eredményre vezet az ev() parancs is:

```
(%i32) ev(F,a=7);
       F;
(%o31) 7 m
(%o32) a m
```

De akár vesszővel is megadhatjuk a behelyettesítést:

```
(%i34) F,a=7;
       F;
(%o33) 7 m
(%o34) a m
```

Kifejtés alkalmazása.

(%i35) expand((a+b)^3);

(%o35) $b^{3}+3\,a\,b^{2}+3\,a^{2}\,b+a^{3}$

(%i36) expand((a+b)··7−(b+2·a)··3);

(%o36) $b^{7}+7\,a\,b^{6}+21\,a^{2}\,b^{5}+35\,a^{3}\,b^{4}+35\,a^{4}\,b^{3}-b^{3}+21\,a^{5}\,b^{2}-6\,a\,b^{2}+7\,a^{6}\,b-12\,a^{2}\,b+a^{7}-8\,a^{3}$

Szorzattá alakítás:

(%i37) factor(a··2+b··2+2·a·b);

(%o37) $(b+a)^{2}$

(%i38) factor(b^3+3·a·b^2+3·a^2·b+a^3);

(%o38) $(b+a)^{3}$

Egyszerűsítés:

(%i39) a/b+c/b+7/b;

(%o39) $\frac{c}{b}+\frac{a}{b}+\frac{7}{b}$

(%i40) ratsimp(a/b+c/b+7/b);

(%o40) $\frac{c+a+7}{b}$

Trigonometrikus kifejezés egyszerűsítése:

(%i41) trigsimp(sin(x)^2+cos(x)^2);

(%o41) 1

Trigonometrikus kifejezés kifejtése:

(%i42) trigexpand(cos(2·x));

(%o42) $\cos(x)^2 - \sin(x)^2$

Tört számlálója és nevezője:

(%i43) num((7+a)/b);

(%o43) $a+7$

(%i44) denom((7+a)/b);

(%o44) b

Változó együtthatója:

(%i45) coeff(45·a+27·b−16·c−100,b);

(%o45) 27

Határérték számítás:

(%i46) limit(sin(x)/x,x,0);

(%o46) 1

Taylor-sor megadása. Első paraméeter a függvény, második a változó, harmadik az érték ami körül akarjuk a sort kifejteni, negyedik pedig a fokszám:

(%i47) taylor(sin(x),x,0,20);

(%o47)/T/ $x-\frac{x^3}{6}+\frac{x^5}{120}-\frac{x^7}{5040}+\frac{x^9}{362880}-\frac{x^{11}}{39916800}+\frac{x^{13}}{6227020800}-\frac{x^{15}}{1307674368000}+\frac{x^{17}}{355687428096000}-\frac{x^{19}}{121645100408832000}+\ldots$

Lista sorbarendezése növekvő és csökkenős sorrendben:

(%i50) lista:[2,3,64,89,1,4,9,0,1]$
sort(lista);
sort(lista,ordergreatp);

(%o49) [0,1,1,2,3,4,9,64,89]

(%o50) [89,64,9,4,3,2,1,1,0]

3 *Mátrixműveletek*

Sorvektor megadása:

(%i51) vek1:matrix([2,0,1]);

(vek1) $\begin{bmatrix} 2 & 0 & 1 \end{bmatrix}$

Oszlopvektor megadása:

(%i52) vek2:matrix([1],[2], [3]);

(vek2) $\begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$

Vektorok skaláris szorzata:

(%i55) v1:[2,3,4]$
v2:[3,−2,−7]$
v1.v2;

(%o55) −28

Vektorok vektoriális szorzata (vect package betöltésével). A vektorszorzás nem értékelődik ki automatikusan, kell az express parancs:

(%i57) load(vect);
express(v1~v2);

(%o56) *C:\maxima−5.40.0\share\maxima\5.40.0\share\vector\vect.mac*

(%o57) [−13,26,−13]

Mátrix magadása:

(%i59) mat1:matrix([1,2,3],[2,0,4],[3,4,1]);
mat2:matrix([1,2,3],[4,5,6],[7,8,9]);

(mat1) $\begin{bmatrix} 1 & 2 & 3 \\ 2 & 0 & 4 \\ 3 & 4 & 1 \end{bmatrix}$

(mat2) $\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$

Egységmátrix megadása:

(%i60) ident(3);

(%o60) $\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$

Diagonális mátrix megadása egyszerűen:

(%i61) diag1:diag_matrix(1,2,3);

0 errors, 0 warnings

(diag1) $\begin{bmatrix} 1 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 3 \end{bmatrix}$

Mátrix sora, oszlopa és adott (sor és oszlop) eleme:

(%i62) row(mat1,3);

(%o62) $\begin{bmatrix} 3 & 4 & 1 \end{bmatrix}$

(%i63) col(mat1,2);

(%o63) $\begin{bmatrix} 2 \\ 0 \\ 4 \end{bmatrix}$

(%i64) mat1[2,3];

(%o64) 4

Alapműveletek:

(%i65) mat1·7;

(%o65) $\begin{bmatrix} 7 & 14 & 21 \\ 14 & 0 & 28 \\ 21 & 28 & 7 \end{bmatrix}$

(%i66) 3·mat1−2·mat2;

(%o66) $\begin{bmatrix} 1 & 2 & 3 \\ -2 & -10 & 0 \\ -5 & -4 & -15 \end{bmatrix}$

(%i67) mat1.mat2;

(%o67) $\begin{bmatrix} 30 & 36 & 42 \\ 30 & 36 & 42 \\ 26 & 34 & 42 \end{bmatrix}$

Transzponált:

(%i68) transpose(mat2);

(%o68) $\begin{bmatrix} 1 & 4 & 7 \\ 2 & 5 & 8 \\ 3 & 6 & 9 \end{bmatrix}$

Trace:

(%i69) mat_trace(mat1);

(%o69) 2

Determináns:

```
(%i71) determinant(mat1);
       determinant(mat2);
```

(%o70) 28
(%o71) 0

Inverz:

```
(%i72) invert(mat1);
```

(%o72) $$\begin{bmatrix} -\frac{4}{7} & \frac{5}{14} & \frac{2}{7} \\ \frac{5}{14} & -\frac{2}{7} & \frac{1}{14} \\ \frac{2}{7} & \frac{1}{14} & -\frac{1}{7} \end{bmatrix}$$

```
(%i73) invert(mat2);
```

expt: undefined: 0 to a negative exponent.
-- an error. To debug this try: debugmode(true);

Látható, hogy a mat2 inverzére hibát ír ki, mivel determinánsa zérus, így inverze nem létezik!

Mátrix rank értéke:

```
(%i74) rank(mat1);
```

(%o74) 3

Karakterisztikus polinom:

```
(%i75) charpoly(mat1,x);
```

(%o75) $(1-x)(-(1-x)x-16)+3(3x+8)-2(2(1-x)-12)$

```
(%i76) ratsimp(%);
```

(%o76) $-x^3+2x^2+28x+28$

Sajátérték számításnál a kimenet tartalmazza a sajátértéket és utána annak multiplicitását. A lenti példánál 3 és -1 a sajátérték és mindekttő 1-szer szerepel:

```
(%i77) eigenvalues(matrix([1,2],[2,1]));
```

(%o77) [[3,−1],[1,1]]

De a következő mátrixnál 1 a sajátérték kétszeres multiplicitással:

```
(%i78) eigenvalues(matrix([1,0],[0,1]));
```

(%o78) [[1],[2]]

A sajátértékeket a part() paranccsal szedhetjük ki a kimenetből. De a part(,1) helyett a first () parancsot is használhatjuk, vagy csak egyszerűen [1]. Például:

```
(%i80) eigenvalues(matrix([1,2],[2,1]));
       part(%,1);
```

(%o79) [[3,−1],[1,1]]
(%o80) [3,−1]

```
(%i81) first(eigenvalues(matrix([1,2],[2,1])));
```

(%o81) [3,−1]

```
(%i83) eigenvalues(matrix([1,2],[2,1]))$
       %[1];
```

(%o83) [3,−1]

Nézzük meg a mat1 sajátértékeit. A mátrix szimmetrikus, tehát tudhatjuk, hogy sajátértékei valósak. Lássuk mit ad a szoftver válaszként:

```
(%i84) ev1:eigenvalues(mat1)[1];
```

(ev1) $$\left[\left(\frac{-1}{2}-\frac{\sqrt{3}\,\%i}{2}\right)\left(\frac{22\sqrt{7}\,\%i}{3}+\frac{638}{27}\right)^{1/3}+\frac{88\left(\frac{\sqrt{3}\,\%i}{2}+\frac{-1}{2}\right)}{9\left(\frac{22\sqrt{7}\,\%i}{3}+\frac{638}{27}\right)^{1/3}}+\frac{2}{3},\left(\frac{\sqrt{3}\,\%i}{2}+\frac{-1}{2}\right)\left(\frac{22\sqrt{7}\,\%i}{3}+\frac{638}{27}\right)^{1/3}+\frac{88\left(\frac{-1}{2}-\frac{\sqrt{3}\,\%i}{2}\right)}{9\left(\frac{22\sqrt{7}\,\%i}{3}+\frac{638}{27}\right)^{1/3}}+\frac{2}{3},\left(\frac{22\sqrt{7}\,\%i}{3}+\frac{638}{27}\right)^{1/3}+\frac{88}{9\left(\frac{22\sqrt{7}\,\%i}{3}+\frac{638}{27}\right)^{1/3}}+\frac{2}{3}\right]$$

Látható, hogy a válasz komplex szám. Próbáljuk egyszerűsíteni:

(%i85) ratsimp(ev1);

(%o85)
$$\left[-\frac{\left(\sqrt{3}\,\%i+1\right)\left(198\sqrt{7}\,\%i+638\right)^{2/3}-4\left(198\sqrt{7}\,\%i+638\right)^{1/3}-88\sqrt{3}\,\%i+88}{6\left(198\sqrt{7}\,\%i+638\right)^{1/3}},\frac{\left(\sqrt{3}\,\%i-1\right)\left(198\sqrt{7}\,\%i+638\right)^{2/3}+4\left(198\sqrt{7}\,\%i+638\right)^{1/3}-88\sqrt{3}\,\%i-88}{6\left(198\sqrt{7}\,\%i+638\right)^{1/3}},\frac{\left(198\sqrt{7}\,\%i+638\right)^{2/3}+2\left(198\sqrt{7}\,\%i+638\right)^{1/3}+88}{3\left(198\sqrt{7}\,\%i+638\right)^{1/3}}\right]$$

(%i86) fullratsimp(ev1);

(%o86)
$$\left[-\frac{\left(\sqrt{3}\,\%i+1\right)\left(198\sqrt{7}\,\%i+638\right)^{2/3}-4\left(198\sqrt{7}\,\%i+638\right)^{1/3}-88\sqrt{3}\,\%i+88}{6\left(198\sqrt{7}\,\%i+638\right)^{1/3}},\frac{\left(\sqrt{3}\,\%i-1\right)\left(198\sqrt{7}\,\%i+638\right)^{2/3}+4\left(198\sqrt{7}\,\%i+638\right)^{1/3}-88\sqrt{3}\,\%i-88}{6\left(198\sqrt{7}\,\%i+638\right)^{1/3}},\frac{\left(198\sqrt{7}\,\%i+638\right)^{2/3}+2\left(198\sqrt{7}\,\%i+638\right)^{1/3}+88}{3\left(198\sqrt{7}\,\%i+638\right)^{1/3}}\right]$$

Nemigazán sikerült, ugyanis nehezen kezeli a képzetes számot tartalmazó kifejezések egyszerűsítését. Azonban ha alkalmazzuk a rectform() parancsot, akkor a komplex számot "a+b·i" alakban igyekszik megadni. Nézzük mi az eredmény:

(%i87) rectform(ev1);

(%o87)
$$\left[\%i\left(\frac{88\left(\frac{\sin\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}}+\frac{\sqrt{3}\cos\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}}\right)}{9}+\frac{(-1)\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}\sin\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2}-\frac{\sqrt{3}\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}\cos\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2}\right)+\frac{88\left(\frac{\sqrt{3}\sin\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}}+\frac{(-1)\cos\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}}\right)}{9}+\frac{\sqrt{3}\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}\sin\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2}+\frac{(-1)\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}\cos\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2}+\frac{2}{3},\right.$$
$$\%i\left(\frac{88\left(\frac{\sin\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}}-\frac{\sqrt{3}\cos\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}}\right)}{9}+\frac{(-1)\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}\sin\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2}+\frac{\sqrt{3}\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}\cos\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2}\right)+\frac{88\left(\frac{(-1)\cos\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}}-\frac{\sqrt{3}\sin\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}}\right)}{9}-\frac{\sqrt{3}\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}\sin\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2}+\frac{(-1)\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}\cos\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{2}+\frac{2}{3},$$
$$\left.\%i\left(\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}\sin\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)-\frac{88\sin\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{9\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}}\right)+\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}\cos\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)+\frac{88\cos\left(\frac{\operatorname{atan}\left(\frac{9\sqrt{7}}{29}\right)}{3}\right)}{9\left(\frac{8\,22^{3/2}}{27}\right)^{1/3}}+\frac{2}{3}\right]$$

Próbálja szimolikusan kezelni. De nézzük mi történik, ha numerikusan nézzük:

(%i88) rectform(float(ev1));

(%o88) $[8.881784197001252\ 10^{-16}\ \%i-1.148244785016935, -4.440892098500626\ 10^{-16}\ \%i-3.608821331886398, 6.757066116903334-1.110223024625157\ 10^{-16}\ \%i]$

Most azt vehetjük észre, hogy a kapott három sajátérték képzetes része nagyon kicsi szám, gyakorlatilag numerikusan zérus. Ez a numerikus sajátértékszámító algoritmus hibája. Ha biztosan tudjuk, hogy szimmetrikus mátrix sajátértékeit kell számolni, akkor célszerű erre saját parancsot írni, hiszen létezik anailitikus zárt alakú kifejezés.

Hasznos lehet még ha nem a beépített függvényt használjuk, hanem behívjuk a lapack (Linear Algebra PACKage) csomagot, majd ezt követően alkalmazzuk a dgeev() parancsot a sajátérték számításra:

(%i89) load(lapack)$

0 errors, 0 warnings
0 errors, 0 warnings
0 errors, 0 warnings
0 errors, 0 warnings
0 errors, 0 warnings
0 errors, 0 warnings
0 errors, 0 warnings
0 errors, 0 warnings
0 errors, 0 warnings

(%i90) dgeev(mat1);

(%o90) [[6.757066116903334, −1.148244785016935, −3.608821331886398], ***false***, ***false***]

A kapott kimeneti változó első eleme tartalmazza a sajátértékeket:

```
(%i91) ev1:dgeev(mat1)[1];
(ev1)  [6.757066116903334,−1.148244785016935,−3.608821331886398]
```

Nézzünk még egy példát az alkalmazásra ahol komplex sajátértékek is vannak:

```
(%i92) mat3:matrix([1,2,3],[3,1,2],[2,3,1]);
```

(mat3) $\begin{bmatrix} 1 & 2 & 3 \\ 3 & 1 & 2 \\ 2 & 3 & 1 \end{bmatrix}$

```
(%i93) ev3:eigenvalues(mat3)[1];
```

(ev3) $[-\frac{\sqrt{3}\,\%i+3}{2},\frac{\sqrt{3}\,\%i-3}{2},6]$

```
(%i94) rectform(float(ev3));
(%o94) [−0.8660254037844386 %i−1.5,0.8660254037844386 %i−1.5,6.0]

(%i95) dgeev(mat3)[1];
(%o95) [6.0,0.8660254037844387 %i−1.5,−0.8660254037844387 %i−1.5]
```

Az értékek ugyanazok, csak nincs ugyanúgy sorbarendezve!

Sajátvektor számításnál a kimenet első eleme a sajátérték számítás eredményét tartalmazza, míg a msáodik elem a hozzájuk tartozó sajátvektorokat:

```
(%i96) eigenvectors(matrix([1,2],[2,1]));
(%o96) [[[3,−1],[1,1]],[[[1,1]],[[1,−1]]]]

(%i98) eival:eigenvectors(matrix([1,2],[2,1]))[1];
       eivec:eigenvectors(matrix([1,2],[2,1]))[2];
(eival) [[3,−1],[1,1]]
(eivec) [[[1,1]],[[1,−1]]]
```

De érdemes megjegyezni, hogy a dgeev() paranccsal is számíthatjuk a sajátvektorokat, ráadásul normalizálva kapjuk meg őket. Például a mat1 mátrix esetén:

```
(%i99) dgeev(mat1,true,false);
```

(%o99) $[[6.757066116903334,-1.148244785016935,-3.608821331886398],\begin{bmatrix} 0.5286440268564944 & 0.8341108351332052 & -0.1574630356062556 \\ 0.5428004299799702 & -0.4747982147655856 & -0.6927729414959655 \\ 0.6526125849863068 & -0.280759274056619 & 0.7037549601592097 \end{bmatrix},false]$

Az első elem a sajátértékeket tartalmazza, míg a második elem a három normalizált sajátvektort. A sajátvektor számítás ellenőrzése:

```
(%i107) eval1:dgeev(mat1,true,false)[1]$
        evec1:dgeev(mat1,true,false)[2]$
        λ1:eval1[1];
        λ2:eval1[2];
        λ3:eval1[3];
        v1:col(evec1,1);
        v2:col(evec1,2);
        v3:col(evec1,3);
(λ1)   6.757066116903334
(λ2)   −1.148244785016935
(λ3)   −3.608821331886398
```

(v1) $\begin{bmatrix} 0.5286440268564944 \\ 0.5428004299799702 \\ 0.6526125849863068 \end{bmatrix}$

(v2) $\begin{bmatrix} 0.8341108351332052 \\ -0.4747982147655856 \\ -0.280759274056619 \end{bmatrix}$

(v3) $\begin{bmatrix} -0.1574630356062556 \\ -0.6927729414959655 \\ 0.7037549601592097 \end{bmatrix}$

```
(%i110) mat1.v1−λ1·v1;
        mat1.v2−λ2·v2;
        mat1.v3−λ3·v3;
```

(%o108) $\begin{bmatrix} 1.332267629550188\ 10^{-15} \\ -1.332267629550188\ 10^{-15} \\ 0.0 \end{bmatrix}$

(%o109) $\begin{bmatrix} -2.220446049250313\ 10^{-16} \\ 1.110223024625157\ 10^{-16} \\ 8.326672684688674\ 10^{-16} \end{bmatrix}$

(%o110) $\begin{bmatrix} 0.0 \\ -4.440892098500626\ 10^{-16} \\ -8.881784197001252\ 10^{-16} \end{bmatrix}$

Látható, hogy az eredmény a zérus vektornak felel meg, az eltérés csak a numerikus számítás hibájából adódik.

4 Egyenletek

Egyenlet megadása:

```
(%i111) egy1:77·x+6=160;
```

(egy1) $77\,x+6=160$

Egyenlet jobb és bal oldala:

```
(%i112) rhs(egy1);
```

(%o112) 160

```
(%i113) lhs(egy1);
```

(%o113) $77\,x+6$

Megoldás:

```
(%i114) megoldas:solve(egy1,x);
```

(megoldas) $[x=2]$

Másik példa több megoldás esetére:

```
(%i115) solve(x·x=5,x);
```

(%o115) $[x=-\sqrt{5},x=\sqrt{5}]$

A megoldás során nem rendelődik hozzá a változóhoz az érték! Jelen esetben az "x" változó értéke továbbra is üres, nem 2:

```
(%i116) x;
```

(%o116) x

Lineáris egyenletrendszer megoldása:

```
(%i119) e1:3·z+2·y+x=7$;
        e2:4·z+2·x=8$;
        e3:z+4·y+3·x=9$;
```

```
(%i120) linsolve([e1,e2,e3],[x,y,z]);
```

(%o120) $[x=\frac{10}{7},y=\frac{6}{7},z=\frac{9}{7}]$

Használhatjuk a linsolve_by_lu() parancsot is, ekkor az m·x=b egyenletrendszert oldjuk meg ahol m az együttható mátrix, x az ismeretlen vektor és b elemei ismertek:

```
(%i122) b:[7,8,9]$
        first(linsolve_by_lu(mat1,b));
```

(%o122) $\begin{bmatrix} \frac{10}{7} \\ \frac{6}{7} \\ \frac{9}{7} \end{bmatrix}$

A find_root() parancssal pedig egyenletek gyökeit kereshetjük:

```
(%i123) find_root(x·x−4=0,x,0,10);
```

(%o123) 2.0

0 és 10 jelenti a határokat amik között keressük a gyököt. Fontos, hogy az előjelek különbözőek legyenek a határokon.

5 Deriválás, integrálás

Deriválás:

(%i124) `diff(sin(x^3),x);`

(%o124) $3\,x^2 \cos\left(x^3\right)$

Többszöri deriválás:

```
(%i126) diff(sin(x^3),x,3);
        ratsimp(%);
```

(%o125) $-54\,x^3 \sin\left(x^3\right) - 27\,x^6 \cos\left(x^3\right) + 6\cos\left(x^3\right)$

(%o126) $\left(6 - 27\,x^6\right)\cos\left(x^3\right) - 54\,x^3 \sin\left(x^3\right)$

Integrálás:

(%i127) `integrate(1/(1+x),x);`

(%o127) $\log\left(x+1\right)$

Határozott integrál:

(%i128) `integrate(1/(1+x),x,1,2);`

(%o128) $\log\left(3\right) - \log\left(2\right)$

Néha az integrálás elvégzéséhez további feltételezések is szükségesek. Az alábbi páldánál a Maxima rékérdez az a paraméter előjelére amire válaszolnunk kell és csak ezt követően adja ki a megoldást:

(%i129) `integrate( 1 / (x^2 + a), x);`

Is a positive or negative? positive;

(%o129) $$\frac{\operatorname{atan}\left(\frac{x}{\sqrt{a}}\right)}{\sqrt{a}}$$

De ezt elérhetjük úgy is, hogy feltételt állítunk be az a paraméterhez. Majd a művelet elvégzése után törölhetjük a feltételt:

```
(%i132) assume(a > 0)$
        integrate( 1 / (x^2 + a), x);
        forget(a > 0)$
```

(%o131) $$\frac{\operatorname{atan}\left(\frac{x}{\sqrt{a}}\right)}{\sqrt{a}}$$

Láncszabályt is ismeri a szoftver

```
(%i137) f(x) := x^2 $
        diff(f(x), x);
        g(y) := sin(y)$
        g(f(x));
        diff( g(f(x)) , x);
```

(%o134) $2\,x$

(%o136) $\sin\left(x^2\right)$

(%o137) $2\,x \cos\left(x^2\right)$

Sok esetben nem létezik zárt alakú kifejezés a határozatlan integrálhoz. Ebben az esetben használhatjuk a numerikus integrálást a határozott integrál számításához:

(%i138) `integrate(sin(cos(x)),x,0,1);`

(%o138) $$\int_0^1 \sin\left(\cos\left(x\right)\right)\,dx$$

(%i139) `romberg(sin(cos(x)),x,0,1);`

(%o139) 0.738643019171958

6 Függvényábrázolás

Célszerű első körben beállítani a Temp könyvtárat, mert sok esetben a Windows default beállításnál nincs jogosultsága a Maximának használni a default Temp könyvtárat. Legyen például C:/Temp.

(%i140) `maxima_tempdir:"C:/Temp";`

(maxima_tempdir) *C:/Temp*

(%i141) wxplot2d(sin(x)·(1−4·exp(0.7·x)/(x^3)), [x,2, 10]);

(%t141) 

(%o141)

Ha több függvényt ábrázolunk akkor automatikusan jelöli őket:

(%i142) wxplot2d([sqrt(x),x··2],[x,0,1]);

(%t142) 

(%o142)

Jobb egérbombbal kattintva el tudjuk menteni a képet. Fontos megjegyezni, hogy lehet felugró gnuplot ablakban is plottoltatni, ahol több lehetőségünk van (pl: PDF-be mentés, nagyítás, ... stb)

(%i143) plot2d([sqrt(x),x··2],[x,0,1]);
(%o143) *[C:/Temp/maxout2648.gnuplot]*

Megadhatjuk a függőleges tartományt is ha korlátozni akarjuk, de ilyenkor levág részeket:

(%i144) wxplot2d(sin(x)·(1−4·exp(0.7·x)/(x^3)), [x,2, 10],[y,−0.5,0.5]);

plot2d: some values were clipped.

(%t144) 

(%o144)

Felirat megadása:

(%i146)
```
wxplot2d(sin(x)·(1−4·exp(0.7·x)/(x^3)), [x,2, 10],[legend,"függvény neve"]);
wxplot2d([sqrt(x),x··2],[x,0,1],[legend,"gyökfüggvény","négyzetes függvény"]);
```

(%t145)


(%o145)

(%t146)


(%o146)

Tengelyfeliratok:

(%i147)
```
wxplot2d(exp(x)/(1.4e4), [x,1, 14],[xlabel,"szorgalmi hét sorszáma"],[ylabel,"Hallgatók terheltsége [%]−ban"]);
```

(%t147)


(%o147)

Logaritmikus tengelyek:

```
(%i149) wxplot2d(exp(x)/(1.4e4), [x,1, 14],[xlabel,"szorgalmi hét sorszáma"],[ylabel,"Hallgatók terheltsége [%]−ban"],logy);
        wxplot2d(exp(x)/(1.4e4), [x,1, 14],[xlabel,"szorgalmi hét sorszáma"],[ylabel,"Hallgatók terheltsége [%]−ban"],logx,logy);
```

(%t148)



(%o148)

(%t149)



(%o149)

Parametrikus függvényábrázolás:

```
(%i150) wxplot2d([parametric, cos(t), sin(t), [t,0,2·%pi]],
            [x, −sqrt(2), sqrt(2)], same_xy)$
```

(%t150)



Diszkrét adatok ábrázolása:

```
(%i151) data:[[1,2],[2,3],[10,0],[15,2],[20,7]]$
```

(%i152) wxplot2d ([discrete, data],[color,red])$

(%t152)



(%i153) wxplot2d ([discrete, data],
[style,points], [point_type,diamond], [color,red])$

(%t153)



3D függvény ábrázolása:

(%i154) wxplot3d(exp(−x^2 − y^2), [x,−2,2],[y,−2,2]);

(%t154)



(%o154)

Ha külön ablakban nyitjuk meg akkor forgatni is tudjuk:

(%i155) plot3d(exp(−x^2 − y^2), [x,−2,2],[y,−2,2]);

(%o155) *[C:/Temp/maxout2648.gnuplot]*

7 Egyebek

- Hasznos az F1 help használata. Tegyük a kurzort a beírt függvényre majd nyomjunk F1-t.
- Esc + karakter + ESC kombinációval görög karaktereket írhatunk gyorsan
- Nagyon hasznos lehet a beépített menük használata, számos függvény és parancs elérhető
- Célszerű sokszor kommentelni. Adott sor végén a /· ·/ jelek közé írhatjuk a kommentet
- Sok esetben a billentyűkombinációk megkönnyítik a munkát. Pl: CTRL+1 esetén text cella beszúrása. A menüsorban láthatóak a parancsok, érdemes megnézni őket
- ..stb